
概要
SolarWinds1のような攻撃は、組織内のアプリケーション
が未知の(悪質な可能性のある)ホストと通信を開始する
のを検出することの難しさを証明しました。そして、その
見落としは、深刻な「サプライチェーン攻撃」につなが
る可能性があります。このホワイトペーパーでは、ネット
ワークデータ、統計データ、JA3/JA3s2ハッシュ (Zeek3

提供)、そしてSplunkを使用して、この問題に対応する方
法について説明します。私たちが行った調査の主な目的
は、ネットワークセキュリティ担当者が、従来なら見落と
していた悪質なアクティビティを検出して攻撃者の優位
に立てるようにすることです。

このホワイトペーパーでは、重要なサーバーで悪質なア
クティビティを検出する方法をいくつか紹介します。現場
の実務担当者に読んでいただきたい内容ですが、「はじ
めに」と「まとめ」のセクションはCISOや管理者の方に
も参考になるでしょう。初めて観測されたデータポイント
や発生頻度の低いデータポイントの検出など、時間をか
けてテストしたクエリーについて詳しく説明します。また、
アノマリ検出の技法を使って、悪質な可能性のあるプロ
セスを特定する方法も説明します。この調査の知見は、
多くの面で、サプライチェーン攻撃を検出するための新
しい手法や技法を求めるセキュリティ担当者に大いに役
立つでしょう。

いずれの知見も、ソフトウェアサプライチェーン攻撃を
検出するための特効薬というわけではありません。それ
でも、検出スピードを上げて攻撃を難しくするためには
有効です(たとえば熊に遭遇したときに、熊より速く走れ
なくても、仲間より速く走れればリスクは減ります)。SOC
の若手アナリストからベテランの脅威ハンターまで、セ
キュリティ担当者が早期に価値を実感できるよう、Zeek
のような簡単に入手して導入できるツールと、すぐに実装
できるクエリーを組み合わせた方法を考えました。私た
ちが目指したのは、いわば、干し草の山を徹底的に小さ
くすることで、その中の針をできるだけ見つけやすくする
ことです。

はじめに
多くのソフトウェア製品は、自動更新、コンテンツサブス
クリプション、データアップロードなどをサポートするため、
バックグラウンドで開発元と通信する「Phone home」
機能を備えています。そして、それらのソフトウェアの多
くは、組織のクラウド環境やオンプレミス環境内の特権的
な場所に導入されます。たとえばSolarwinds Orionは、
ネットワーク監視というその主要な用途をサポートするた
めに、通常、ネットワークアクセスの点で完全な自由が
与えられます。同様に、人気の高いコードカバレッジツー
ルのCodecovは、継続的インテグレーションパイプライ
ンに組み込まれ、ソースコード、パスワード、APIキー、
証明書署名鍵などに自由にアクセスできます。一般的に、
ソフトウェアの開発元が、Phone homeトラフィックの正
規の通信先に使用するIPアドレス、ドメイン名、証明書
を公開することはありません。公開するベンダーもありま
すが、多くの場合は顧客から開示要求があった場合のみ
で、情報自体も頻繁に変更されます。攻撃者は、こうし
た状況を踏まえて、この仕組みを積極的に悪用します。
開発元のシステムに侵入し、製品のコードを改ざんして、
正規の「Phone home」機能を装ってユーザーの機密デー
タを別の場所にリダイレクトさせるのです。

ベンダーには当然、ソフトウェアサプライチェーンを構成
する自社システムを保護する責任がありますが、ユーザー
側でも、悪質なPhone homeアクティビティを検出するた
めに何かできないでしょうか。多くのエキスパートが勧め
るのが、正常なネットワークアクティビティのベースライ
ンを設定し、そこから外れるアクティビティを検出したら
アラートを出すという方法です。しかしこれは、口で言う
ほどたやすいことではありません。IPアドレスは頻繁に変
わり、数分のうちに、多くの場合は別のユーザーに割り
当て直されます。

この問題の解決策の1つが、高精度なデータポイントを
収集して異常なアクティビティを検出することです。今回
の調査を始めるにあたって、私たちは、意図的に目標を
狭め、多くの制限を設けることで、設定やインフラの変
更が最小限または皆無で済み、多くの読者が実践しやす
いように心掛けました。大切なのは「今後」ではなく
「今すぐ」使えることです。「今後」では遅すぎるかもしれ
ないのです。収集対象として一般的で広くサポートされ
ている高精度データポイントの1つがJA3ハッシュとJA3s
ハッシュです。ここではまとめて「JA3/s」と呼びます。
このホワイトペーパーでは、JA3/sハッシュを高精度デー
タポイントとして活用し、Splunkのコア機能を使って異
常なアクティビティを突き止めます。

1. https://www.fireeye.com/blog/threat-research/2020/12/evasive-
attacker-leverages-solarwinds-supply-chain-compromises-with-sunburst-
backdoor.html

2. https://engineering.salesforce.com/tls-fingerprinting-with-ja3-and-
ja3s-24736285596

3. https://zeek.org/

JA3/sとは
JA34は、SSL/TLSハンドシェイクプロセスで使用する特
定のMD5ハッシュ値を生成するためのオープンソースの
手法です。クライアントのハンドシェイクリクエストに含ま
れる主な属性がセッションから抽出され、連結されて、
その文字列からMD5アルゴリズムによってハッシュが
生成されます。具体的にはクライアント側の次の属性が
抽出されます。

SSLVersion,Cipher,SSLExtension,EllipticCurve, 

EllipticCurvePointFormat

これらの値を連結し、その結果からハッシュを計算するこ
とにより、リクエストを行った特定のクライアントやライブ
ラリ/バイナリの一貫したハッシュを生成できます。たとえ
ば、Trickbotに感染したコンピューターからのセッション
をこの方法で分析すると、TrickbotのバイナリのJA3
ハッシュは、6734f37431670b3ab4292b8f60f29984
になります。このハッシュは、Trickbotのバイナリからリ
クエストされたSSL/TLSセッションであれば、送信元と
宛先のIPアドレスが変わっても常に同じです。攻撃者に
とって、IPアドレスやドメイン名を変えることは比較的容
易ですが、別のSSL/TLSライブラリを使用するようにマル
ウェアを改変することは容易ではありません。そのため、
JA3を監視すれば、攻撃者に対してネットワーク接続を
隠ぺいするハードルを上げることができます。

さらに、同様の方法でサーバーセッションのJA3ハッシュ
を計算できます。これは正式には「JA3sハッシュ」と
言います。JA3ハッシュの生成プロセスは同じですが、
サーバーセッションから抽出する主な属性が多少異なり
ます。これは、クライアントのリクエストによってサーバー
の応答が異なることがあるためです。具体的にはサー
バー側の次の属性が抽出されます。

SSLVersion,Cipher,SSLExtension

JA3とJA3sのいずれも、さまざまなツールを使って簡単
にネットワークトラフィックから取得できます。しかし、
今回の調査の目標を踏まえて、ここではハッシュ生成
ツールとしてZeekを使用します。

検出
今回の調査では、異常なアクティビティを検出する複数
の方法を開発しました。目標は、Splunk Enterprise
以外のコンポーネントを使用する必要がほとんどまたは
全くないシンプルなクエリーを構築することでした。その
結果、いくつかの注意事項と制限が生じたので、まずは
それらをお伝えします。

注意事項と制限 
サプライチェーン攻撃に限らず、いずれのタイプの悪質な
アクティビティでも、100%確実に検出する方法はありま
せん。そのため、私たちの目標は常に、利用可能なツー
ルを使って、異常なアクティビティをできるだけ多く突き
止めることです。実際の企業データ5とテスト環境で生成
したデータ6を用いたテストでは、異常なJA3/sハッシュ
から異常なアクティビティを高い確率で検出できることが
わかりました。とはいえ、実際の効果は多くの要因に左
右されます。誤検知の件数を減らすには、許可リストの
作成はほぼ必須です。また、この調査ではJA3/sハッシュ
を使って異常なアクティビティを検出することに重点を
置いているため、SSL/TLSで暗号化されていないネット
ワーク接続には効果がありません。

さらに、これらの方法は組織内の適切なネットワーク
セグメントを対象に使用する必要があります。クエリーで
は、アウトバウンド接続をする内部ホストが検出されます。
そのため、組織内からのWebブラウズを仲介するホスト
や、SSL/TLSセッション経由でさまざまな外部サービス
と頻繁に通信するホストには効果がありません。これら
のクエリーは、クライアントとしてアウトバウンド接続する
ことがあまりない内部ホストやネットブロックを対象に使
用してください。

SSL/TLSインターセプトやインスペクションを使用する場
合も効果がありません。SSL/TLSインターセプトを使用
すると、リクエスト側のクライアントには、実際の外部サー
バーの属性とは異なる属性が提示されます。そのため、
JA3/sハッシュに基づいて異常なアクティビティを検出す
ることは実質的に不可能です。同僚はこれを、私の名前
を取って「LaFerreraパラドックス」と呼びます(私にとっ
ては迷惑な話ですが)。一般的な検出手法ではサプライ
チェーン攻撃を検出できないことがわかるくらいの知識
があるからこそ、それらの手法を無効にしなければ使え
ない緩和策を講じることになる、という皮肉です。

4. https://github.com/salesforce/ja3

5. これらのクエリーを実際のデータで試すため、Splunkの複数のお客様にご協
力いただきましたことを感謝申し上げます。このご協力がなければ、今回の調
査ははるかに困難なものになっていたでしょう。

6. https://github.com/mlaferrera/SEC1745/code

異常なアクティビティの検出
今回の調査を通じて、私たちはさまざまなアプローチで検出方法を開発しました。初めて観測されたアクティビティや
発生頻度の低いアクティビティの検出のような従来の技法はもちろん、Splunk Enterpriseのanomalydetection7コマ
ンド(頻度分析を用いてJA3sハッシュのようなカテゴリフィールドの通常ではありえない異常な値を検出するSPLコマン
ド)を使用するアプローチや、ルックアップ8とSPL9を使用する類似のアプローチなど、最新の戦略も含まれます。前述の
とおり、多くの場合は、正常なネットワークトラフィックをサーチ結果から除外するための許可リストを作成する必要があ
ります。

クエリー
私たちは、大半のセキュリティ担当者が最小限のスキルや変更ですぐに導入できるシンプルな方法を探りました。また、
幅広いデータソースで効果が実証されているクエリーを使用することも重視しました。いずれのクエリーも、悪質なアク
ティビティを100%確実に検出できるわけではありません。それでも、私たちの経験から言えば、目前の問題を緩和する
最善の道は、シンプルで一定の効果のある解決策から始めることです。

以下のクエリーはすべて、異常なネットワークトラフィックを検出するためのもので、すでに実績があります。前述の制
限があることはご留意ください。また、これらのクエリーを使用するときは常に、ネットワークアドレスを実際の環境に
合わせて変更する必要があります。今回の調査と以下のクエリーの最新版は、GitHubリポジトリ10で公開しています。では、
各クエリーの概要とSplunkでの使い方を順に説明していきます。

初めて観測されたアクティビティの検出(First Seenクエリー)11

First Seenクエリーでは、ネットワークアクティビティを詳細に把握していて、許可リストを使用する場合に、最も効果的
に異常なアクティビティを検出できます。また、結果は特定の期間に限定されるため、指定する期間によって結果が
大きく異なります。期間が極端に長いか短いと、異常なアクティビティが見落とされたり、正常なトラフィックと区別でき
なかったりすることがあります。調査では、多くの場合、期間を7日に設定すると最適な結果が得られ、ターゲットにし
た悪質なアクティビティがサーチ結果の上位20件に入りました。また、下の図にはありませんが、一般的なJA3sハッシュ
とserver_nameまたはそのいずれかを記載した許可リストを追加して既知のエンティティを除外すると、精度が向上し
ます。

sourcetype="bro:ssl:json" ja3="*" ja3s="*" src_ip IN (192.168.70.0/24) 

| stats earliest(_time) as earliest latest(_time) as latest by ja3, ja3s, src_ip, server_name 

| eval maxlatest=now() 

| eval isOutlier=if(earliest >= relative_time(maxlatest, "-1d@d"), 1, 0) 

| table ja3, ja3s, src_ip, server_name, earliest, latest, maxlatest, isOutlier 

| convert ctime(earliest) ctime(latest) ctime(maxlatest) 

| sort earliest desc

7. https://docs.splunk.com/Documentation/Splunk/latest/SearchReference/Anomalydetection.

8. https://docs.splunk.com/Documentation/Splunk/8.2.2/Knowledge/Aboutlookupsandfieldactions

9. https://www.splunk.com/ja_jp/resources/search-processing-language.html

10. https://github.com/mlaferrera/SEC1745

11. https://github.com/mlaferrera/SEC1745/blob/main/queries/firstseen.txt

発生頻度の低いアクティビティの検出(Rarest)12

発生頻度の低い、server_nameごとのJA3sハッシュを検出するには、許可リストの作成がほぼ欠かせません。作成し
なくても、既知の悪質なホストがサーチ結果の上位20件に入ることはありますが、常に検出できるとは限りません。結
果は特定の期間に限定されるため、指定する期間によって結果が大きく異なります。期間が極端に長いか短いと、悪質
な通信の頻度や期間によっては結果が偏ることがあります。そのため、このクエリーは、ここで紹介する他の方法の補
足として使用するのが効果的です。

sourcetype="bro:ssl:json" ja3="*" ja3s="*" src_ip IN (192.168.70.0/24) 

| eventstats count as total 

| stats values(ja3), values(dest_ip), values(src_ip) values(total) as total count by server_

name ja3s 

| eval perc=round((count/total)*100,4) 

| sort + perc

アノマリ検出13 

First SeenクエリーとRarestクエリーで不審なアクティビティを確認したら、anomalydetectionのヒストグラム機能
を使用します。Splunkのこのネイティブコマンドは、データから異常なイベントを検出するのに役立ちます。結果に含ま
れる各イベントの発生確率を計算して、発生確率が異常に低いイベントを特定します。この方法は、クエリーで指定し
た期間内で異常なイベントを検出したいときに便利です。ただし、悪質なイベントの発生頻度が正常なトラフィックと似
ている場合、それらが正常なアクティビティに見えることがある点に注意してください。

テストでは、結果を調整して適切な結果のみに絞り込むために、確率のしきい値(pthresh)を変更する必要がありまし
た。その際、最も効果が高かったpthresh値は0.001でした。しかし実際に使用する場合は、収集するデータ量と、
異常なイベントをどの程度の感度で検出したいかによって、この値を独自に調整する必要があるでしょう。

anomalydetectionコマンドを使用する方法は、24～ 48時間の期間で異常なアクティビティを検出する場合に高い
効果を発揮します。期間がそれよりも長いと、効果が下がります。テストでは、サブネットマスク「/24」のネットブロッ
ク1つで構成される小規模ネットワークの場合、許可リストなしでも、悪質なアクティビティが一貫して上位30件に含ま
れました。一方、複数のネットブロックを含むネットワークや、ネットブロックの範囲が広いネットワークでは、あまり
一貫した結果は得られませんでした。既知の悪質なアクティビティを検出することはできても、上位30件に入らないこ
とがありました。この場合、悪質なアクティビティが上位30件に入るようにするには、正常なホストを記載した許可リス
トを指定するのが効果的です。

12. https://github.com/mlaferrera/SEC1745/blob/main/queries/rarest.txt.

13. https://github.com/mlaferrera/SEC1745/blob/main/queries/anomalydetection.txt

sourcetype="bro:ssl:json" ja3="*" ja3s="*" src_ip IN (192.168.70.0/24) 

| anomalydetection method=histogram action=annotate pthresh=0.0001 src_ip, ja3, ja3s 

| stats sparkline max(log_event_prob) AS "Max Prob", min(log_event_prob) AS "Min Prob", 

values(probable_cause) AS "Probable Causes", values(dest_ip) AS "Dest IPs", values(server_

name) AS "Server Names", values(ja3) AS "JA3", values(src_ip) as "Source IPs" count by ja3s 

| table "Server Names", "Probable Causes", "Max Prob", "Min Prob", "Dest IPs", ja3s, "JA3", 

"Source IPs", count 

| sort "Min Prob" asc

ルックアップによるアノマリ検出14、15、16

今回の調査では、SPLのanomalydetectionコマンドを再現して、結果をルックアップテーブルに保存することで、サー
チを拡張する方法も開発しました。このクエリーでは、イベントのsrc_ip、ja3、ja3sを一組として同様の発生確率
を計算し、outputlookupコマンドを使って結果をCSV形式のルックアップテーブルに保存します。

sourcetype="bro:ssl:json" ja3="*" ja3s="*" src_ip IN (192.168.70.0/24) 

| eval id=md5(src_ip+ja3+ja3s) 

| stats count by id,ja3,ja3s,src_ip 

| eventstats sum(count) as total_host_count by src_ip,ja3 

| eval hash_pair_likelihood=exact(count/total_host_count) 

| sort src_ip ja3 hash_pair_likelihood 

| streamstats sum(hash_pair_likelihood) as cumulative_likelihood by src_ip,ja3 

| eval log_cumulative_like=log(cumulative_likelihood) 

| eval log_hash_pair_like=log(hash_pair_likelihood) 

| outputlookup hash_count_by_host_baselines.csv

14. https://github.com/mlaferrera/SEC1745/blob/main/queries/outputlookup.txt

15. https://github.com/mlaferrera/SEC1745/blob/main/queries/inputlookup.txt

16. https://github.com/mlaferrera/SEC1745/blob/main/queries/outputlookup-update.txt

ルックアップテーブルを生成したら、lookupコマンドを使ったクエリーを実行して、異常なアクティビティを検出します。
lookupコマンドによる2つ目のクエリーで指定する期間は、outputlookupを使用する1つ目のクエリーの対象期間に
収まる範囲にするのが最適です。テストでは、outputlookupで過去7日間のデータからテーブルを生成した後、過去
最大48時間の範囲で異常なイベントを検索しました。

sourcetype="bro:ssl:json" ja3="*" ja3s="*" src_ip IN (192.168.70.0/24) 

| eval id=md5(src_ip+ja3+ja3s) 

| lookup hash_count_by_host_baselines.csv id as id OUTPUT count, total_host_count,log_cumulative_

like, log_hash_pair_like 

| table _time, src_ip, ja3s, server_name, subject, issuer, dest_ip, ja3, log_cumulative_like, 

log_hash_pair_like, count, total_host_count 

| sort log_hash_pair_like

最後に、常に最新の確率を得るため、ルックアップテーブルの情報を最新にするためのクエリーを実行します。このク
エリーは、1つ目のoutputlookupクエリーを少し変更するだけで作成できます。たとえば、元のoutputlookupクエ
リーでは過去7日間を対象としましたが、ここでは、クエリーを24時間ごとに実行して、過去24時間分のデータを対象
に含めます。元のクエリーの結果を追加し、前回のクエリーが完了した後で開始するように期間を制限します。

sourcetype="bro:ssl:json" ja3="*" ja3s="*" src_ip IN (192.168.70.0/24) 

| eval id=md5(src_ip+ja3+ja3s) 

| stats count by id,ja3,ja3s,src_ip 

| append  

    [| inputlookup hash_count_by_host_baselines.csv] 

| stats sum(count) as count by id,ja3,ja3s,src_ip 

| eventstats sum(count) as total_host_count by src_ip,ja3 

| eval hash_pair_likelihood=exact(count/total_host_count) 

| sort src_ip ja3 hash_pair_likelihood 

| streamstats sum(hash_pair_likelihood) as cumulative_likelihood by src_ip,ja3 

| eval log_cumulative_like=log(cumulative_likelihood) 

| eval log_hash_pair_like=log(hash_pair_likelihood) 

| outputlookup hash_count_by_host_baselines.csv

この方法の結果は、効果の点でも検索時間の点でも、anomalydetectionコマンドを使用する方法と比べて遜色あり
ません。ただし一般的に、毎日使用する場合では、lookupを使ったクエリーよりもanomalydetectionの方が約100倍
高速であることがテストでわかりました。また、対象となるネットワークが拡大したときは許可リストも必須です。テスト
では、許可リストを指定すると、既知の悪質なアクティビティが一貫して上位30件に含まれました。

JA3sとSysmon17,18

ネットワークデータとローカルプロセスの実行データの組み合わせは、脅威検出において非常に価値のあるデータソー
スです。SplunkでSysmon19データを収集すれば、外部と通信しているプロセスを特定して、JA3/sハッシュと相関付け
ることができます。これにより、Windowsプロセス、JA3/sハッシュ、server_nameを結び付けることができます。たと
えば、外部ホストに接続するpowershell.exeプロセスを検出できます。この関連データを収集するには、Network 
Connection Initiatedイベント(イベントコード3)を収集するようにSysmonを設定します。Olaf Hartong氏20は、Sysmon
をモジュール式で設定するためのユーティリティを開発し、オープンソース化しています。これを使用すると、必要なデー
タをすばやく収集できます。

私たちは、問題の確認後にJA3sとSysmonの結果を相関付けるための2つの方法を開発しました。1つ目が次に示す
方法で、SysmonデータとJA3/ネットワークデータを検索しますが、より効率的なSplunkデータモデル21は使用してい
ません。

(source="XmlWinEventLog:Microsoft-Windows-Sysmon/Operational" EventCode=3 src_ip IN 

(192.168.70.0/24)) 

 OR 

(sourcetype="bro:ssl:json" ja3=* ja3s=*) 

| eval src_ip=if(sourcetype == "bro:ssl:json",'id.orig_h','src_ip') 

| eval src_port=if(sourcetype == "bro:ssl:json",'id.orig_p','src_port') 

| eval dest_ip=if(sourcetype == "bro:ssl:json",'id.resp_h','dest_ip') 

| eval dest_port=if(sourcetype == "bro:ssl:json",'id.resp_p','dest_port') 

| stats values(ja3) as ja3 values(ja3s) as ja3s values(process_path) as process_path 

values(server_name) as server_name by src_ip dest_ip dest_port 

| search ja3=* ja3s=* process_path=* NOT process_path IN ("<unknown process>")

17. https://github.com/mlaferrera/SEC1745/blob/main/queries/sysmon-simple.txt

18. https://github.com/mlaferrera/SEC1745/blob/main/queries/sysmon-multisearch.txt

19. https://docs.microsoft.com/en-us/sysinternals/downloads/Sysmon

20. https://github.com/olafhartong/Sysmon-modular

21. https://docs.splunk.com/Documentation/Splunk/latest/Knowledge/Aboutdatamodels

2つ目の方法は、データモデルを使用してパフォーマンスを高めています。いずれの方法も結果は同じです。ただし、
テストでは、データモデルを使用するクエリーの方が約4倍高速でした。

| multisearch 

 [ from datamodel:Network_Traffic.All_Traffic 

 | search sourcetype="xmlwineventlog" source="XmlWinEventLog:Microsoft-Windows-Sysmon/

Operational" src_ip IN (192.168.70.0/24) 

 | rename app as process_path] 

 [ search sourcetype="bro:ssl:json" ja3=* ja3s=*] 

| eval src_ip=if(sourcetype == "bro:ssl:json",'id.orig_h','src_ip') 

| eval src_port=if(sourcetype == "bro:ssl:json",'id.orig_p','src_port') 

| eval dest_ip=if(sourcetype == "bro:ssl:json",'id.resp_h','dest_ip') 

| eval dest_port=if(sourcetype == "bro:ssl:json",'id.resp_p','dest_port') 

| stats count values(ja3) as ja3 values(ja3s) as ja3s values(process_path) as process_path, 

values(server_name) as server_name by src_ip dest_ip dest_port 

| search ja3=* ja3s=* process_path=* NOT process_path IN ("<unknown process>")

環境によっては、これらのクエリーで、悪質な可能性のあるアクティビティのトリアージはできても、異常なアクティビティ
の特定はできない場合があります。テストでは、前述の方法で悪質な可能性のある異常なアクティビティを特定してから、
JA3sとSysmonを使用するクエリーを使ってイベントをトリアージするという使い方が最も効果的でした。また、
server_nameを取得できる場合はその値が含まれるため、異常なアクティビティや不審なアクティビティを手動で特定
する場合も役立つでしょう。

まとめ
JA3/sを使用して悪質なアクティビティを検出するのは、
決して完璧な方法ではありません。いくつかの注意事項
や制限を受け入れる必要があります。それでも、制限内
でデータを収集して分析すれば、他の方法では特定で
きない悪質なアクティビティを検出できる可能性があり
ます。

JA3/sハッシュの生成方法を考えると、悪質なアクティビ
ティを高い信頼度で検出するのには問題があります。しか
し、アクセスが制限された重要なネットワークセグメント
やホストで異常なアクティビティを検出する分には、詳細
調査をすべきと判断できるレベルの信頼度を確保できま
す。私たちは今回の調査を通じて、ネットワークセンサー
のデータセットで一般的なデータを活用するためのシンプ
ルで新しい方法を探りました。SSL/TLSフィンガープリン
ティングの技法をより有効に活用する方法は、当然、ほ
かにもあるでしょう。この調査の知見が、組織で実行でき
る対策をより深く理解し、さらに高度な対策を探求するた
めに役立てば幸いです。これらの知見を最大限に活用す
るために、ここで紹介した方法をぜひ実際に試してみてく
ださい。こちらのデータは、https://bots.splunk.com
のインタラクティブなSplunkワークショップでご利用いた
だけます。

要点
このホワイトペーパーでは、Splunkコマンドやクエリー
を使って異常なSSL/TLS通信を検出する方法を紹介しま
した。これらのクエリーが実際の環境で効果を発揮する
かどうかは、さまざまな要因に左右されます。どのクエリー
も、多くの場合、最適に機能させるには、ある程度の調
整や変更が必要です。また、SSL/TLSイベントを大量に
生成するサーバー /ホストには効果がありません。一定
の効果が保証されるのは、クライアントとして外部と頻
繁に通信することのない、重要度の高いホストやネット
ブロックに限られます。

SURGeアラートにぜひご登録ください

多くの環境では、anomalydetectionコマンドを使用
する方法が効果的ですが、この方法は拡張性の面で限界
があります。拡張性が必要な場合は、ルックアップによる
アノマリ検出の方法が妥当で効率的かもしれません。
さらに、サーチ結果の件数を絞り込み、悪質なアクティ
ビティをより確実に検出するには、承認済みの証明書、
ドメイン、JA3/sハッシュを記載した許可リストを作成す
ることが欠かせません。

